Ball on Inclined Plane



Figure 1: Ball-on-inclined-plane with sensors, an actuator (servo motor) and a PCB

1.) Understand the Physical Principles and the Mathematical Analyses



Figure2: Model of the Inclined Plane

Table 1: Characteristics values of the physical model

Name	Value
d	105 mm
h	19 mm
l = $\mathbf{a_2}$	118.72 mm
r = $\mathbf{a_1}$	18 mm

c = $\mathbf{b_2}$	36.22 mm
$\mathbf{h_k}$	17 mm
$\mathbf{l_k}$	117.5 mm

$$\boxed{\gamma = \tan^{-1}(h/d)}$$

$$\boxed{\delta = \tan^{-1}(\mathbf{h_k}/\mathbf{l_k})}$$

Law of cosine for two triangles:

$$c_1^2 = a_1^2 + b_1^2 - 2a_1b_1\cos\gamma_1$$

Where,

$$\gamma_1 = \pi - (\beta + \gamma)$$

$$\therefore c_1 = \sqrt{a_1^2 + b_1^2 - 2a_1b_1\cos\gamma_1}$$

$$a_1^2 = c_1^2 + b_1^2 - 2c_1b_1\cos\alpha_1$$

$$\boxed{\alpha_1 = \cos^{-1}\left(\frac{c_1^2 + b_1^2 - a_1^2}{2a_2c_1}\right)}$$

$$b_2^2 = a_2^2 + c_1^2 - 2a_2c_1\cos\beta_2$$

$$\boxed{\therefore \beta_2 = \cos^{-1}\left(\frac{a_2^2 + c_1^2 - b_2^2}{2c_1b_1}\right)}$$

Sum of the angles at pivot point of the plane:

$$\alpha + \delta + \gamma = \alpha_1 + \beta_2$$

Therefore,

$$\boxed{\alpha = \alpha_1 + \beta_2 - \delta - \gamma}$$

$$\boldsymbol{\alpha = \mathbf{f}(\beta)}$$

First Approximation:

$$\boldsymbol{\alpha = (a1/a2)\ \beta}$$

Table 2: Notations

α	alpha
β	beta
γ	gama
δ	delta

2.) Build a SCILAB script for acquiring the relationship between α and β (with β limited to ± 90 degree) and present the relationship in a SCILAB plot.

- Two vector constructs for angle β
    - First case: $\beta \geq (-\gamma)$
    - Second case: $\beta < (-\gamma)$
- Find the right equations for both vectors
- Combine the two cases in one plot α = f(β)
- Obtain the limits for angle β from the plot when the angle α is limited to ±5 degrees
- Plot the first linear approximation in the same plot

Note*: Scilab code written in boxes*

```
//Defining the whole code in terms of a function for ease of operation
function [] = boip() //Start of Function
clc
// Given Values
d = 105e-3;
h = 19e-3;
a2 = 118.72e-3;
a1 = 18e-3;
b2 = 36.22e-3;
hk = 17e-3;
lk = 117.5e-3;
b1 = sqrt(d^2 + h^2);
gama = atand(h/d);
delta = atand(hk/lk);
// Every angle is defined in degrees and the appropriate trigonometric
functions are used
n= 1000; // Number of Steps/Iterations
// Defining different ranges of Beta for which alpha1 would be positive or
negative in the domain -90 to +90 Degrees
B_1 = linspace(-90,-gama,n);
B_2 = linspace(-gama,90,n);
```

Governing Equations in form of Functions

```
function [alpha] = equation1(beta) // Equation for beta < - gama

    gama1 = 180 - beta - gama;
    c1 = sqrt(a1.^2 + b1.^2 - 2*a1.*b1.*cosd(gama1));
    alpha1 = acosd((c1.^2+b1.^2-a1.^2)./(2*c1.*b1));
    beta2 = acosd((a2.^2+c1.^2-b2.^2)./(2*a2.*c1))
    alpha = -alpha1 + beta2 - gama - delta;
endfunction

function [alpha] = equation2(beta) // Equation for beta >= -gama

    gama1 = 180 - beta - gama ;
    c1 = sqrt(a1.^2 + b1.^2 - 2*a1.*b1.*cosd(gama1));
    alpha1 = acosd((c1.^2+b1.^2-a1.^2)./(2*c1.*b1));
    beta2 = acosd((a2.^2+c1.^2-b2.^2)./(2*a2.*c1))
    alpha = alpha1 + beta2 - gama - delta;
endfunction
```

First Case: $\beta \geq (-\gamma)$



Figure 3: Beta is greater than - gama

Alpha for $\beta \geq -\gamma$

$$\boxed{\alpha = \alpha_1 + \beta_2 - \delta - \gamma}$$

```
// Beta >= - gama
A_2 = equation2(B_2); //Using equation2 function here because it is defined
for B >= -gama
figure(2)
plot(B_2,A_2)
xlabel('Beta (Degrees)',"fontsize",3)
ylabel('Alpha (Degrees)',"fontsize",3)
title('Beta ranging from -gama to +90',"fontsize",4)
```



Figure 4: Alpha when Beta Ranging from -gama to 90 degrees

Second Case: β < (−γ)



Figure 5: Beta less than -gama

Alpha for $\beta < -\gamma$

$$\boxed{\alpha = -\alpha_1 + \beta_2 - \delta - \gamma}$$

```
// Beta < - gama
A_1 = equation1(B_1); //Using equation1 function here because it is defined for B < -gama
figure(1)
plot(B_1,A_1)
xlabel('Beta (Degrees)',"fontsize",3)
ylabel('Alpha (Degrees)',"fontsize",3)
title('Beta ranging from -90 to -gama',"fontsize",4)
```



Figure 6: : Alpha when Beta Ranging from -90 to -gama degrees

Combine the two cases in one plot $\alpha = f(\beta)$

```
// Beta ranges from -90 to +90 Degrees
A1 = [A_1 A_2]; // Concatenating Values of Alpha over the range -90 to +90 Degrees
B1 = [B_1 B_2]; // Concatenating Values of Beta over the range -90 to +90 Degrees
figure(3)
plot(B1,A1)
xlabel('Beta (Degrees)',"fontsize",3)
ylabel('Alpha (Degrees)',"fontsize",3)
title('Beta ranging from -90 to +90 Degrees',"fontsize",4)
```



Figure 7: Alpha when Beta Ranging from -90 to + 90 degrees

Obtain the limits for angle β from the plot when the angle α is limited to ± 5 degrees
Plot the first linear approximation in the same plot

```
// Limits for angle Beta when range of Alpha is: -5 >= alpha <= +5
findbeta = find([A1 >= -5] & [A1 <= 5]); // Defining the limits of Alpha in find function
betalim = B1(findbeta);
beta_max = max(betalim)
beta_min = min(betalim)

mprintf('Lower limit for Beta :\ Beta(Minimum)=%f',beta_min)
mprintf('\n')
mprintf('Upper limit for Beta :\ Beta(Maximum)=%f',beta_max)
```

```
Scilab 6.0.2 Console

Lower limit for Beta : Beta(Minimum)=-35.720358
Upper limit for Beta : Beta(Maximum)=34.502481

-->
```

Figure 8: Output for Maximum and Minimum Values of Beta in the range
-5 ≤Alpha (α) ≤ +5 in Scilab Console

Lower Limit for Beta, Beta (min) = -35.720358 degrees
Upper Limit for Beta, Beta (max) = 34.502481 degrees



Figure 9: First Linear Approximation of Alpha

3.) Answer the following questions:

- **Explain with reasons why the approximation is not so good if angle β is greater than 40 degrees or smaller than -40 degrees.**
- **In your SCILAB simulation, maximize the range of the angle β to ±300 degrees, plot, and explain your output graph.**
- **What are the mechanical problems when the black rod is too long or too short? (with plots and arguments)**

Analysis of First Linear Approximation

Equation:

$$\alpha = (a1/a2)\ \beta$$

```
// Analysing the first Linear Approximation
alpha_linear = (a1/a2)*B1; // First Linear Approximation Equation
figure(4)
plot(B1,A1)
plot(B1,alpha_linear)
leg = legend("Curve: No Approximation", "Straight Line: First Approximation")
xlabel('Beta (Degrees)',"fontsize",3)
ylabel('Alpha (Degrees)',"fontsize",3)
title('Analysing the First Linear Approximation',"fontsize",4)
```



Figure 10: Analyzing the First Linear Approximation of Alpha

Result of the Analysis:

- The first approximation doesn't hold good for Beta(β) < -40 degrees and Beta(β) > +40 degrees because the deviation from the true characteristics becomes too big to ignore.

- In the above graph, apart from the -40<Beta(β)<+40 range the difference between the assumed straight line and the actual curve increases as Beta increases in magnitude.

- Another observation from the above graph is that when the value of Beta(β) increases further from around 70 degrees, the first approximation shows that the value of Alpha is also increasing. But in reality, the value of alpha starts decreasing after reaching a maximum value. This discrepancy in the approximation is not worth accommodating.

Angle β ranges from -300 to +300

```
// Maximizing Range of Beta to - 300 to +300 degrees
// Defining different ranges of Beta for which alpha1 would be positive or negative
in the domain -300 to +300
B300_1 = linspace(-300,-(gama+180),n);
B300_2 = linspace(-(180 + gama),-gama,n);
B300_3 = linspace(-gama,180-gama,n);
B300_4 = linspace((180-gama),300,n);

A300_1 = equation1(B300_1);
A300_2 = equation2(B300_2);
A300_3 = equation1(B300_3);
A300_4 = equation2(B300_4);

// Concatenating Values of Beta over the range -300 to +300 Degrees
B300 = [B300_1 B300_2 B300_3 B300_4];

// Concatenating Values of Alpha over the range -300 to +300 Degrees
A300 = [A300_1 A300_2 A300_3 A300_4];
figure(5)
plot(B300,A300)
xlabel('Beta (Degrees)',"fontsize",3)
ylabel('Alpha (Degrees)',"fontsize",3)
title('Maximising Range of Beta to - 300 to +300 degrees',"fontsize",4)
```



Figure 11: Maximizing Range of Beta to -300 to +300 Degrees

Explanation of the Output Graph:

- As expected, the output graph is Sinusoidal.
- As we can see the graph has a repetitive nature after a 360 Degree rotation of the servomotor. Which is what one would expect because the servomotor arm reaches its original position.
- For a 360 Degree rotation, there is a single value of Alpha (α) for two values of Beta (β). The value of Alpha (α) will be same at two positions of motor arm below or above the horizontal plane.
- When the motor arm is reaching its highest or lowest position in the vertical plane, the change in Alpha (α) due to change in Beta (β) becomes very small. This can be also seen from the changing slope of curve.
- For the rest of the positions of the servomotor arm, the change in Alpha (α) is very drastic due to change in Beta (β).

Mechanical problem when the black rod(b_2) is too long or too short

1) When the Black Rod gets Too Short



Figure 12 a, b, c, d & e: Varying the length of Black Rod below the given Value of 36.22 mm

Result of the Analysis:

- From the above plots we can clearly see that descreasing the Length of the Black Rod (b_2) has a direct impact on the result value of Alpha(α) i.e. Angle of the inclined plane.
- As the length of black rod (b_2) decreases more and more, the value of Alpha(α) also decreases. This means that inclined plain will lean towards left (in reference to the Figure 1) more and more for the change in angle Beta(β).
- This means the angle Alpha(α) will be negative for most of the values of angle Beta(β).
- And after a certain point as shown in Figure 12 d, the angle Alpha(α) will be negative only.
- After reaching this state, the balancing of the ball on the inclined plane will become impossible.
- Decreasing the length of Black Rod (b_2) from around less than 30mm will result into a mechanism that can't make a complete rotation.
- Decreasing the length of the Black rod (b_2) further will result in the mechanical failure of the four bar mechanism and the whole setup will become absolutely disfunctional.

2) When the Rod is Too Long



Figure 13 a, b, c, d & e: Varying the length of Black Rod above the given Value of 36.22 mm

Result of the Analysis:

- As we can see from the above graphs, increasing the Length of the Black Rod (b_2) from the given value results in increase of the positive range of angle Alpha(α) i.e. Angle of the inclined plane.
- This means that the inclined planed will lean more towards right (in reference of the Figure 1).
- This means the angle Alpha(α) will be positive for most of the values of angle Beta(β).
- And after a certain point as shown in Figure 13 e, the angle Alpha(α) will be positive only.
- After reaching this state, the balancing of the ball on the inclined plane will become impossible.
- Increasing the length of the Black rod (b_2) further will result in the mechanical failure of the four bar mechanism and the whole setup will become absolutely disfunctional.
